

干预时间序列分析与程序包的比较

王群勇（南开大学数量经济研究所）

2023年8月14号，哈尔滨商业大学

内容

干预模型简介

干预模型: itsa

干预模型: titsa

干预时间序列模型

在时间序列回归模型中，当解释变量为政策干预变量时，即得到干预模型。

ITSA的核心任务是检验变量的趋势（确定趋势或随机趋势）是否受由于干预变量发生了改变。

基本的两类干预：脉冲（pulse），阶梯（step）

- pulse function: $pulse_t = 1(t = 0), v_t(t = 0, 1, \dots)$.
- step function: $step_t = 1(t \geq 0)$


$step_t = step_{t-1} + pulse_t$, or $step_t = (1 - B)^{-1}pulse_t$.

ITSA广泛应用于公共政策评估（新疫苗有效性、安全带法，交通限速等）或紧急情况的影响（金融危机等）。

干预时间序列模型

ITS的基本任务是将政策前的趋势外推得到政策后的反事实。


例：澳大利亚1972年1月开始实施安全带法。对交通死亡人数的影响：



干预时间序列模型

ITS也可以估计处理效应随时间的变化特征，模型中可以体现变量自身的**动态变化规律**。

ITS也适用于多个不同时点的不同干预的情形。



干预时间序列模型

定义干预变量 $D = 1(T \geq \tau)$ 。

即时效应 (immediate effect) :

$$y = \alpha_0 + \alpha_1 t + \delta_0 D + u$$

即时效应+持续效应 (immediate effect + sustained effect) :

$$y = \alpha_0 + \alpha_1 t + \delta_0 D + \delta_1 t \times D + u$$

持续效应:

$$y = \begin{cases} \alpha_0 + \alpha_1(t - \tau) + u, & t < \tau \\ \beta_0 + \beta_1(t - \tau) + u, & t \geq \tau \end{cases}$$

约束 $\alpha_0 = \beta_0$ (piecewise regression, jointpoint regression),

$$\begin{aligned} y &= \alpha_0 + \alpha_1(t - \tau) + \delta_1(t - \tau) \times D + u \\ &= \alpha_0 + \alpha_1(t - \tau)(1 - D) + \beta_1(t - \tau) \times D + u \\ &= \beta_0 + \alpha_1 t + \delta_1(t - \tau) \times D + u \end{aligned}$$

例：安全带法

澳大利亚1972年1月开始实施安全带法，考察其对交通死亡人数的影响。

(McCleary et al., 2017, sec 5.2.1; McDowall et al. 2019, sec 4.2.1):

数据：fatalityaus.dta, 1950Q1-1974Q4的季度数据。

Box-Cox transformation: $y^* = (y^\lambda - 1)/\lambda$,

例：安全带法

```
. use fatalityaus, clear  
. itsa y , single trperiod(1972q1) lag(1)
```

Time variable: qdate, 1950q1 to 1976q4
Delta: 1 quarter

```
Regression with Newey-West standard errors      Number of obs      =      108  
Maximum lag = 1                                F( 3,      104)    =      247.83  
                                                Prob > F          =      0.0000
```

	Coefficient	Newey-West std. err.	t	P> t	[95% conf. interval]	
_t	.0740647	.0031212	23.73	0.000	.0678752	.0802542
_x1972q1	-.6180887	.3062339	-2.02	0.046	-1.225362	-.0108153
_x_t1972q1	-.0797835	.0357192	-2.23	0.028	-.1506159	-.0089511
_cons	3.89452	.1766139	22.05	0.000	3.544288	4.244752

例：安全带法

```
. des

Contains data from fatalityaus.dta
Observations:      108
Variables:         10          14 Jan 2023 21:57
```

Variable name	Storage type	Display format	Value label	Variable label
qdate	float	%tq		
fatality	int	%8.0g		
seatbelt	float	%9.0g		
y	float	%9.0g		y=sqrt(fatality)
ycopy	float	%9.0g		
_y	float	%9.0g		y=sqrt(fatality)
_t	float	%9.0g		
_x1972q1	float	%9.0g		
_x_t1972q1	float	%9.0g		
_s_y_pred	float	%9.0g		Fitted values

例：安全带法

```
. actest
```

```
Cumby-Huizinga test for autocorrelation (Breusch-Godfrey)
```

```
H0: variable is MA process up to order q
```

```
HA: serial correlation present at specified lags >q
```

```
-----
```

H0: q=0 (serially uncorrelated)					H0: q=specified lag-1				
HA: s.c. present at range specified					HA: s.c. present at lag specified				
lags	chi2	df	p-val	lag	chi2	df	p-val		
1 - 1	0.002	1	0.9614	1	0.002	1	0.9614		


```
-----
```

```
Test allows predetermined regressors/instruments
```

```
Test requires conditional homoskedasticity
```

例：安全带法

```
capture drop yhat
predict yhat, xb
ewrapper _x1972q1 _x_t1972q1
capture drop yhatc
predict yhatc, xb
twoway (scatter y qdate, msize(small)) ///
      (tsline yhat if tin(,1971q4), lp(solid) lwidth(thin)) ///
      (tsline yhat if tin(1972q1,), lp(solid) lwidth(medium)) ///
      (tsline yhatc if tin(1972q1,), lp(dash) lwidth(thick)), ///
      tline(1972q1, lp(dash)) legend(off) xtitle("")
```



例：Stata itsa

itsa 估计如下模型：

$$y_t = \alpha_0 + \alpha_1 t + \delta_0 D_t + \delta_1 D_t \times t + u_t, u_t = \rho u_{t-1} + v_t$$

- Newey稳健标准误差；Prais-Winsten转换

(1) 只能估计线性趋势模型，不能估计非线性趋势模型。

(2) 只允许模型误差项为AR(1)过程，而不能是一般的ARMA(p, q)过程。这是由于 itsa 实际上是 newey 和 prais 指令的外壳，而这两个指令仅估计AR(1)模型。

(3) 只允许最简单的干预实验设计，即干预变量为Step类型，不允许其它形式的干预，比如ABAB型设计、或者reversed treatment design等。

(4) 只能估计确定性趋势模型，不能估计随机趋势（含有单位根的过程）模型。

二次趋势的sustained effect

前面的处理方法使得回归函数连续但不可导。如果要保证回归函数连续且可导，要引入二阶多项式函数。

二次多项式函数的连续与可导：

$$m(x) = \begin{cases} \alpha_0 + \alpha_1(x - \tau) + \alpha_2(x - \tau)^2, & x < \tau \\ \beta_0 + \beta_1(x - \tau) + \beta_2(x - \tau)^2, & x \geq \tau \end{cases}$$

当且仅当 $\alpha_0 = \beta_0$ 时， $m(x)$ 在 $x = \tau$ 处是连续的。当 $\alpha_1 = \beta_1$ 时，模型的一阶导数是连续的。

这时模型可以写为：


$$m(x) = \beta_0 + \beta_1 x + \beta_2 x^2 + \beta_3 (x - \tau)^2 1(x \geq \tau).$$

其中， $1(x \geq \tau)$ 为虚拟变量，即括号内的条件如果成立，取值为1；否则，取值为0。

二次趋势的sustained effect

```
use fatalityaus, clear
gen t=_n
gen dum = t>=89
gen ct = t-89
reg y t c.t#c.t c.ct#c.ct#i.dum
predict yf
twoway (scatter y t) (line yf t if d==0, sort(t) lwidth(thick)) ///
      (line yf t if d==1, sort(t) lwidth(thick)) , ///
      xline(89, lp(dash)) legend(off)
```

二次趋势的sustained effect



高阶趋势的sustained effect

类似地，具有连续的一阶导数和二阶导数的分段三次函数可以写为

$$m(x) = \beta_0 + \beta_1 x + \beta_2 x^2 + \beta_3 x^3 + \beta_4 (x - \tau)^3 1(x \geq \tau).$$

一般地，具有 K 个结点 $(\tau_1, \tau_2, \dots, \tau_K)$ 的 p 阶函数为


$$m(x) = \beta_0 + \sum_{j=1}^p \beta_j x^j + \sum_{k=1}^K \gamma_k (x - \tau_k)^p 1(x \geq \tau_k).$$

在干预时序分析中，结点即政策的实施时间，往往是外生给定的。

多个结点的sustained effect

如果样本分为多段，不同的分段由结点连接起来。如果对每一段单独回归模型的话，不同段的参数估计量（包括截距项和斜率项）都是不一样的。

如何使得每一段的回归线是连续的呢？



多个结点的sustained effect

存在多个结点情况下，如果约束回归线在结点 τ_j 处是连续的，那么在该结点建立中心化的趋势变量与该结点对应的虚拟变量乘积。

比如，有两个结点 (τ_1, τ_2) ，回归线在 (τ_1, τ_2) 处连续。那么模型设定为


$$y_t = c + \beta x_t + \delta_1 D_{1t}(x_t - \tau_1) + \delta_2 D_{2t}(x_t - \tau_2) + a_t.$$

其中， $D_{1t} = 1(x_t \geq \tau_1)$ ， $D_{2t} = 1(x_t \geq \tau_2)$ 。

一般地，具有 K 个结点 $(\tau_1, \tau_2, \dots, \tau_K)$ 的线性（一阶）样条函数为

$$y_t = c + \beta x_t + \sum_{k=1}^K \delta_k (x_t - \tau_k) 1(x_t \geq \tau_k) + a_t.$$

多个结点的sustained effect




允许结点处不连续

如果允许回归线在某个结点，比如 τ_j ，不连续，那么模型中加入该结点对应的虚拟变量即可：

$$y_t = c + \beta x_t + \sum_{k=1}^K \delta_k (x_t - \tau_k) \mathbf{1}(x_t \geq \tau_k) + \gamma_j \mathbf{1}(x_t \geq \tau_j) + a_t.$$

```
use fatalityaus, clear
gen t=_n
mkspline v1 13 v2 89 v3 = t // spline function given knots
gen d = t>=13
reg y v1 v2 v3 d
predict yf
twoway (scatter y t) (line yf t if d==0, sort(t) lwidth(thick)) ///
(line yf t if d==1, sort(t) lwidth(thick)) , ///
xline(13 89, lp(dash)) legend(off)
```

允许结点处不连续



内容

干预模型：确定性趋势

干预模型：基于ARIMA转移函数

- **转移函数**

转移函数

线性滤子:

$$y_t = \sum_{i=0}^{\infty} v_i x_{t-i} = v(B)x_t.$$

系数 $\sum_i |v_i| < \infty$.

转移函数模型: $y_t = v(B)x_t + u_t$, 其中

$$V(B) = \frac{w_0 + w_1 B + \cdots + w_s B^s}{1 - \delta_1 B - \cdots - \delta_r B^r}.$$

转移函数

滞后效应:

$$y_t = v(B)x_{t-l} + u_t = \frac{w(B)}{\delta(B)}x_{t-l} + u_t.$$

其中, $\delta(B) = 1 - \delta_1 B - \delta_2 B^2 - \dots - \delta_r B^r$,

由 $v(B)B^l = w(B)/\delta(B)$, $v(B)\delta(B) = w(B)B^{-l}$,

$$v_j - \delta_1 v_{j-1} - \delta_2 v_{j-2} - \dots - \delta_r v_{j-r} = \begin{cases} w_{j-l} & j = l + 1, l + 2, \dots, l + s \\ 0 & j > l + s \end{cases}$$

转移函数

例:

$$y_t = \frac{1.2 - 1.5B + 0.6B^2}{1 - 0.6B} x_t$$

其中 $w(B) = 1.2 - 1.5B + 0.6B^2$, $\delta(B) = 1 - 0.6B$, then

$$\delta(B)^{-1} = 1 + 0.6B + 0.36B^2 + 0.216B^3 + \dots,$$

$$v(B) = 1.2 - 0.78B + 0.132B^2 + 0.0792B^3 + \dots$$

滞后两期,

$$\begin{aligned} y_t &= \frac{1.2 - 1.5B + 0.6B^2}{1 - 0.6B} x_{t-2} \\ &= (1.2B^2 - 0.78B^3 + 0.132B^4 + 0.0792B^5 + \dots) x_t \end{aligned}$$

转移函数

```
tranfunc num (expr) den (expr) , [ lag (#) order (10) ivtype ("step"|"pulse")  
tfcoef (expr) tfcov (expr) ]
```

其中, `num` 和 `den` 分别为分子和分母的多项式。`num` 和 `den` 可以含有参数, 参数的具体数值由 `tfcoef` 设置, `tfcov` 设置参数估计量的方差。

`transfunc` 将转移系数保存在 `r(tf)` 中。如果设置 `tfcov`, 那么 `r(tf)` 为转移系数及其标准差。

转移函数

```
. transfunc num(1 - 1.5*B + 0.6*B^2) den(1 - 0.6*B)
Numerator polynomial:      1 - 1.5*B + 0.6*B^2
Denominator polynomial:    1 - 0.6*B
Lag:                        0
```

order	Coef
0	1
1	-.9
2	.06
3	.036
4	.0216
5	.01296
6	.007776
7	.0046656
8	.0027994
9	.0016796
10	.0010078

转移函数

```
. transfunc num(1 - 0.8*B) den(1 - 1.2*B + 0.5*B^2), lag(2)
Numerator polynomial:          1 - 0.8*B
Denominator polynomial:       1 - 1.2*B + 0.5*B^2
Lag:                           2
```


order	Coef
0	0
1	0
2	1
3	.4
4	-.02
5	-.224
6	-.2588
7	-.19856
8	-.108872
9	-.0313664
10	.0167963

转移函数

```
. transfunc num("1-a*B") den("1+c*B") , tfcoef("{a -> 0.6, c -> 0.5}") tfcov("{0.1, 0.05}, {0.05, 0.008}")  
Numerator polynomial:      1-a*B  
Denominator polynomial:    1+c*B  
Lag:                        0
```

order	Coef	Se
0	1	0
1	-1.1	.4560702
2	.55	.3542316
3	-.275	.2336236
4	.1375	.1437272
5	-.06875	.0849706
6	.034375	.0489339
7	-.0171875	.0276572
8	.0085938	.015412
9	-.0042969	.0084934
10	.0021484	.0046389

```
. transfunc num("1-a*B") den("1+c*B") , ///  
  tfcoef("{a -> 0.6, c -> 0.5}") ///  
  tfcov("{0.1, 0.05}, {0.05, 0.008}")  
. svmat r(tf), names(tf)  
. gen low=tf1-2*tf2  
. gen upp=tf1+2*tf2  
. gen order = _n - 1 in 1/11  
. twoway (rspike upp low order in 2/11) //  
  (connected tf1 order in 1/11) , legend(off)
```



转移函数

定义:

$$S_t = 1(t > T), I_t = 1(t = T),$$

$$S_t = (1 - B)^{-1} I_t.$$

$$w_0 S_t$$

$$\frac{w_0}{1 - \delta_1 B} S_t$$

$$\frac{w_0}{1 - B} S_t$$











$$\left(\frac{w_0}{1 - \delta_1 B} + \frac{w_0}{1 - B} S_t \right) S_t$$

$$w_0 I_t = w_0(1 - B) S_t$$

$$\frac{w_0}{1 - \delta_1 B} I_t = \frac{w_0(1 - B)}{1 - \delta_1 B} S_t$$

$$\frac{w_0}{1 - B} I_t = w_0 S_t$$

$$\left(\frac{w_0}{1 - \delta_1 B} + \frac{w_0}{1 - B} \right) I_t$$

	Step input	Pulse input
$\frac{\omega(B)}{\delta(B)}$		
	Output response	Output response
ω_0		
$\frac{\omega_0}{1 - \delta_1 B}$		
$\frac{\omega_0}{1 - B}$		
$\frac{\omega_0}{1 - \delta_1 B} + \frac{\omega_1}{1 - B}$		

转移函数的ML和NLS估计

设误差项为ARIMA(p, d, q)过程, $\phi(B)u_t = \theta(B)e_t$, 即

$$y_t = \frac{w(B)}{\delta(B)} x_{t-l} + \frac{\theta(B)}{\phi(B)} e_t.$$

上面模型可以等价地写为ARIMAX模型。

多个干预变量,

$$y_t = \sum_{j=1}^m \frac{w_j(B)}{\delta_j(B)} x_{j,t-l_j} + \frac{\theta(B)}{\phi(B)} e_t.$$

NLS估计: $\sum_{t=1}^T e_t^2$, 或者对 $u_t = y_t = \sum_{j=1}^m \frac{w_j(B)}{\delta_j(B)} x_{j,t-l_j}$ 建立ARIMA模型。

内容

干预模型：确定性趋势

干预模型：基于ARIMA转移函数

- 转移函数
- **干预模型**

转移函数

定义阶梯干预变量: $I_t = 1(t > \tau)$, 或脉冲干预变量 $I_t = 1(t = \tau)$ 。

观测变量 y_t 与 I_t 的关系通过转移函数 (transfer function) 联系起来:

$$y_t = f(I_t) + u_t$$

其中, u_t 为 ARMA(p,q) 过程。

zero-order transfer function: $f(I_t) = \omega I_t$

- 脉冲响应函数为: $\omega, 0, 0, \dots$; 阶梯响应函数为: $\omega, \omega, \omega, \dots$.

first-order transfer function: $f(I_t) = \frac{\omega}{1-\delta B} I_t$

- 脉冲响应函数为: $\omega, \omega\delta, \omega\delta^2, \dots$, 收敛于 0.
- 阶梯响应函数为: $\omega, \omega(1 + \delta), \omega(1 + \delta + \delta^2), \dots$, 收敛于 $\omega/(1 - \delta)$.


常见的干预效应

McCleary et al. (2017, Figure 5.1) 设 I_t 为Step干预变量，上面对应的转移函数依次为：

$$\frac{wI_t}{1 - \delta B} I_t$$

$$\frac{w}{1 - B} I_t$$

$$\frac{w(1 - B)}{1 - \delta B} I_t.$$



model:

$$y_t = c + \frac{w_1(B)}{\delta_1(B)} x_{1t} + \frac{w_2(B)}{\delta_2(B)} x_{2t} + \cdots + \frac{\theta(B)}{\phi(B)} \epsilon_t$$

titsa *varlist* [*if*] [*in*], `xnum` (*string*) `xden` (*string*) `enum` (*string*) `eden` (*string*) *options*

其中, `xnum` : $(w_1(B), w_2(B), \dots)$; `xden` : $(\delta_1(B), \delta_2(B), \dots)$; `xnum` 与 `xden` 设定的表达式个数应与 `varlist` 中解释变量的个数相同。

`enum` : $\theta(B)$; `eden` : $\phi(B)$ 。

所有多项式的默认值为1。

options: `inits("{values}")` .

example: `inits("{1, -0.5, 0.2, 0.4}")`

note: 常数项自动加入到模型中。

Stata titsa

例:

$$y_t = c + \frac{w_0 + w_1 B}{1 - \delta B - \delta_2 B^2} x_{1t} + \frac{w_2}{1 - \delta_3 B} x_{2t} + \frac{1 + \theta B}{1 - \phi B} \epsilon_t$$

```
titsa ..., xnum("{w0+w1*B, w2}") xden("{1-delta1*B-delta2*B^2, 1-delta3*B}") enum("1+th*B") eden("1-phi*B")
```

滞后算子的符号可以自行定义，比如"B"或"L"。

例：休息与劳动生产率

Franke and Kaul (1978): 80个周的实验, 33rd-45th周每天工作期间休息25分钟。

数据文件: rbprd.dta (McDowall, 2019, sec 4.1.1)

$$prd_t = w_0 + w_1 I_t + \frac{1 + \theta_1 B}{1 - B} a_t.$$

一段时间之后取消干预, 叫做Removed or Reversed Treatment Designs。这种设计具有更高的内部有效性。

类似的实验设计: 美国1965年强制摩托车驾驶戴头盔, 1975年取消改法律。

例：休息与劳动生产率

```
. use rbprd, clear
. gen t=_n
. tsset t, generic
. titsa prd rb , xnum("{w0+w1*B}") xden("{1}") enum("1+th*B") eden("1-B")
```

Interrupted time series analysis (transfer function):

Number of coef	=	3	Number of obs	=	80
AIC	=	0.9658	R-squared	=	0.1526
BIC	=	1.0551	RSS	=	187.6570
HQIC	=	1.0016	Root MSE	=	2.4371
BG	=	19.6445	PierceBox	=	0.0001
P(BG)	=	0.0015	P(PierceBox)	=	1.0000
LjungBox	=	0.0001	DW	=	2.0037
P(LjungBox)	=	1.0000			

prd	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
w0	.0999991	81.27759	0.00	0.999	-159.2011	159.4011
th	.0000469	.0932403	0.00	1.000	-.1827007	.1827945
const	-37847.03	395.966	-95.58	0.000	-38623.11	-37070.95

$prd_t = c + (w_0 + w_1 B)I_t + \frac{1+\theta_1 B}{1-B} a_t$ 可以等价地写为ARIMA(0,1,1)-X模型:

$$\Delta prd_t = (w_0 + w_1 B)\Delta I_t + (1 + \theta_1 B)a_t.$$

不是所有的干预模型都能等价地表达为线性ARIMA模型的形式。比如,

$$y_t = c + \omega I_t + \frac{1}{1 - \phi B} a_t.$$

两边同时乘以 $1 - \phi B$:

$$(1 - \phi B)y_t = c(1 - \phi B) + \omega(1 - \phi B)I_t + a_t.$$

即

$$\begin{aligned} y_t &= c(1 - \phi) + \phi y_{t-1} + \omega I_t - \omega \phi I_{t-1} + a_t \\ &= \mu + \phi y_{t-1} + \omega I_t + \beta I_{t-1} + a_t. \end{aligned}$$

无约束的ARIMA模型包含 $(\mu, \phi, \omega, \beta)$ 四个参数, 干预模型包含 (c, ω, ϕ) 三个参数。因此, 干预模型是受约束的ARIMA模型, 非线性约束为 $\beta = -\omega\phi$ 。

例：ABAB干预

美沙酮疗法 (methadone maintenance) 的效果(Kirby, Kerwin, Carpendo, Rosenwasser, & Gardener, 2008)。1st-25th: 无干预。

26th-54th: 多种疗法 (包括戒除可卡因, treatment A) ;

55th-73rd: 只戒除可卡因;(treatment B); 74th-91nd: treatment A; 92nd-105th: treatment B.

因变量为尿液的清洁度 (可卡因的含量) 。

数据文件: abab.dta (McDowall, 2019, sec 4.1.4)

$$clean_t = w_0 + w_1 IA_t + w_2 IB_t + \frac{1 + \theta B}{1 - \delta B} a_t.$$

其中, $IA_t = 1(26 \leq t \leq 54 \text{ or } 74 \leq t < 91)$,

$IB_t = 1(55 \leq t \leq 73 \text{ or } 92 \leq t \leq 105)$.

例：ABAB干预

```
. use abab, clear
. gen t=_n
. tsset t, generic
. gen iv = ia+ib
. titsa clean ia ib , xnum("{wa, wb}") xden("{1, 1}") enum("1 + th*B") eden("1-delta*B")
```

Interrupted time series analysis (transfer function):

Number of coef	=	5	Number of obs	=	105
AIC	=	4.6182	R-squared	=	0.6009
BIC	=	4.7446	RSS	=	9210.5544
HQIC	=	4.6694	Root MSE	=	92.1055
BG	=	4.2533	PierceBox	=	4.9077
P(BG)	=	0.5135	P(PierceBox)	=	0.4272
LjungBox	=	5.2025	DW	=	2.0337
P(LjungBox)	=	0.3917			

clean	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
wa	13.04309	3.679267	3.55	0.000	5.831863	20.25432
wb	16.5494	3.967112	4.17	0.000	8.774002	24.32479
th	-.2741823	.3039798	-0.90	0.367	-.8699717	.3216071
delta	.5478122	.2632835	2.08	0.037	.0317861	1.063838
const	21.21477	1.907285	11.12	0.000	17.47656	24.95298

例

澳大利亚1972年1月开始实施安全带法(McCleary et al., 2017, sec 5.2.1; McDowall et al. 2019, sec 4.2.1), 对交通死亡人数的影响:

$$fatality_t^{1/2} = \frac{w_0}{1-B} + \frac{\omega_1}{1-\delta_1 B} I_t + \frac{1+\theta_1 B}{1-B} a_t.$$

等价地表示为

$$(1-\delta_1 B)\Delta fatality_t^{1/2} = w_0(1-\delta_1 B) + \omega_1 \Delta I_t + (1-\delta_1 B)(1+\theta_1 B)a_t.$$

其中, 对常数项和MA的系数是有约束的。无约束模型包含5个参数, 即 $(1, \Delta fatality_{t-1}^{1/2}, \Delta I_t$ 以及MA(1), MA(2)。但ITS模型包含4个参数。

数据文件: 1950Q1 - 1976Q4

```
. use fatalityaus, clear
. gen lny = ln(y)
. titsa lny t seatbelt, xnum("{a, w0}") xden("{1, 1-delta*B}") enum("1 + th*B") eden("1")
```


Interrupted time series analysis (transfer function):

Number of coef	=	5	Number of obs	=	108
AIC	=	-2.7257	R-squared	=	0.9247
BIC	=	-2.6015	RSS	=	6.1499
HQIC	=	-2.6753	Root MSE	=	0.0597
BG	=	11.2255	PierceBox	=	7.3899
P(BG)	=	0.0471	P(PierceBox)	=	0.1932
LjungBox	=	7.8006	DW	=	1.9839
P(LjungBox)	=	0.1676			

lny	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
a	.0218988	.0011795	18.57	0.000	.0195871	.0242105
w0	-.0808747	.0434273	-1.86	0.063	-.1659907	.0042413
delta	.8839765	.0961277	9.20	0.000	.6955696	1.072383
th	.1648418	.0977366	1.69	0.092	-.0267185	.356402
const	2.859439	.0524887	54.48	0.000	2.756563	2.962315

例


```
. matlist e(V)  
. transfunc num("w0") den("1-delta*B") , ///  
  tfcoef("{w0 -> -0.08, delta ->0.88 }") ///  
  tfcov("{0.0019, 0.0039}, {0.0039, 0.0092}") ///  
  ivtype("step")  
. svmat r(tf), names(tf)  
. gen low=tf1-2*tf2  
. gen upp=tf1+2*tf2  
. gen order = _n - 1 in 1/11  
. twoway (rspike upp low order in 2/11) (connected tf1 order in 1/11) , legend(off)
```



复杂影响

比如,

$$y_t = w_0 I_t + \frac{w_1}{1 - \delta B} \Delta I_t + a_t = \left(w_0 + \frac{w_1(1 - B)}{1 - \delta B} \right) I_t + a_t.$$




复杂影响

醉酒合法化的影响：（数据：
decrimdrunk.dta, 1966M1-
1975M12）。美国1969年1月开始，
醉酒合法化，相应的逮捕案件下降。

模型：

```
. use decrimdrunk, clear  
. gen lnarrest=ln(arrest)  
. tsline lnarrest, tline(1969m1 1975m12, lp(dash))
```



$$\ln(\text{arrest}_t) = \theta_0 + w_1 I_t + w_2 \Delta I_t + \frac{1}{1 - \phi_1 B - \phi_2 B^2} a_t.$$

```
. gen diffx=d.decrim
. titsa lnarrest decrim diffx , xnum("{w1, w2}") xden("{1, 1-delta*B}") enum("1") eden("1 - phi1*B - phi2*B^2")
```

Number of coef	=	6	Number of obs	=	119
AIC	=	-3.0761	R-squared	=	0.9617
BIC	=	-2.9360	RSS	=	4.7135
HQIC	=	-3.0192	Root MSE	=	0.0417
BG	=	3.5434	PierceBox	=	3.5124
P(BG)	=	0.6168	P(PierceBox)	=	0.6215
LjungBox	=	3.6776	DW	=	1.8838
P(LjungBox)	=	0.5967			

lnarrest	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
w1	-1.429592	.0795391	-17.97	0.000	-1.585486	-1.273698
w2	-.846525	.2096062	-4.04	0.000	-1.257346	-.4357044
delta	.5971083	.19213	3.11	0.002	.2205405	.9736761
phi1	.7420003	.0905779	8.19	0.000	.5644709	.9195297
phi2	-.270768	.093133	-2.91	0.004	-.4533054	-.0882306
const	7.966084	.03507	227.15	0.000	7.897349	8.03482


复杂影响

```
. matlist e(V)
. transfunc num("w1*(1-delta*B)+w2") den("1-delta*B"), tfcoef("{w1 -> -1.43, w2 ->-0.85, delta->0.6}") tfcov("{0.0063, -0.0041}, {-0.0041,0.0439}")
. svmat r(tf), names(tf)
. gen low=tf1-2*tf2
. gen upp=tf1+2*tf2
. gen order = _n - 1 in 1/11
. twoway (rspike upp low order in 2/11) //
  (connected tf1 order in 1/11) , legend(off)
```

家庭法对离婚率的影响 (McCleary et al. 2017, sec 5.4.1)。(数据: ausdivorce.dta, 1901-2009)。澳大利亚1975年开始Family Law Act, 允许无过失离婚。利用1946-2009期间的数据估计如下模型, 并计算政策的干预效应:

$$divorce_t = c + w_1 I_t + \frac{w_2}{1 - \delta B} \Delta I_t + \frac{1}{(1 - B)(1 - \phi B)} a_t.$$

```
use ausdivorce, clear  
tsline divorce , tline(1975, lp(dash))
```



复杂影响

```
. gen law = year>=1975
. gen diffx=D.law
. titsa divorce law diffx if inrange(year,1946,2009), xnum("{w1, w2}") xden("{1, 1 - delta*B}") enum("1 + th*B") eden("(1 - B)*(1 - phi*B)")
```

```
Number of coef =          6          Number of obs =          64
AIC =          15.1707          R-squared =          0.9961
BIC =          15.3731          RSS =          1.8645e+08
HQIC =          15.2505          Root MSE = 3214644.3445
BG =          7.7656          PierceBox =          9.0477
P(BG) =          0.1696          P(PierceBox) =          0.1072
LjungBox =          9.8124          DW =          1.5872
P(LjungBox) =          0.0807
```

divorce	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
w1	36775	2885.738	12.74	0.000	31119.06	42430.95
w2	-30034.97	2093.247	-14.35	0.000	-34137.66	-25932.28
delta	-.3246358	.0446525	-7.27	0.000	-.4121531	-.2371186
th	-1.177461	.0145871	-80.72	0.000	-1.206051	-1.14887
phi	.8362787	.0766464	10.91	0.000	.6860545	.986503
const	7441.81	255.9174	29.08	0.000	6940.221	7943.399

复杂影响

```
. matlist e(V)
. transfunc num("w1*(1-delta*B)+w2") den("1-delta*B"), tfcoef("{w1 ->36775, w2 ->-30034, delta->-0.32}") tfcov(".....")
. svmat r(tf), names(tf)
. gen low=tf1-2*tf2
. gen upp=tf1+2*tf2
. gen order = _n - 1 in 1/11
. twoway (rspike upp low order in 2/11) //
  (connected tf1 order in 1/11) , legend(off)
```

itsa的扩展

多组比较 (did)

合成控制法 (synthetic control)

程序安装

运行 `titsa` 需要安装Stata程序包 `whereis` 和 `Wolframscript.exe`。

安装Wolframscript:

- 方式1: 直接安装Mathematica。
- 方式2: 单独安装 `wolframscript.exe`。下载:
<https://www.wolfram.com/wolframscript/> (免费注册账号)

安装 `whereis` 。

```
ssc install whereis, replace
```

第三步, 在 `whereis` 指定 `wolframscript.exe` 的路径。比如,

```
whereis wolframscript "C:\Program Files\Wolfram Research\Mathematica\13.2\wolframscript.exe"
```

James Lopez Bernal, Steven Cummins, Antonio Gasparrini, Interrupted time series regression for the evaluation of public health interventions: a tutorial, *International Journal of Epidemiology*, Volume 46, Issue 1, February 2017, Pages 348–355.

Dennis J , Ramsay T, Turgeon AF, Zarychanski R. Helmet legislation and admissions to hospital for cycling related head injuries in Canadian provinces and territories: interrupted time series analysis. *BMJ*2013;346:f2674

Grundy C , Steinbach R, Edwards P, Green J, Armstrong B, Wilkinson P. Effect of 20 mph traffic speed zones on road injuries in London, 1986-2006: controlled interrupted time series analysis. *BMJ*2009;339:b4469.

Lopez Bernal J , Gasparrini A, Artundo C, McKee M. The effect of the late 2000s financial crisis on suicides in Spain: an interrupted time series analysis. *EJPH*. 2013;23(5):732–6.

Kontopantelis E , Doran T, Springate DA, Buchan I, Reeves D. Regression based quasi-experimental approach when randomisation is not an option: interrupted time series analysis. *BMJ*2015;350:h2750.

Linden A , Adams JL. Applying a propensity score-based weighting model to